

Created By

AHMED HABIB

1

بسم الله الرحمن الرحيم

(وما أسألكم عليه من اجر إن اجري إلا علي الله)

اسأل الله العلي العظيم أن يوفقني ويوفقكم الي مافيه الخير لي ولكم وأرجومن كل من قرأ الكتاب ان لا يبخل علي بالدعاء لي ولوالدي وان يغفر لي عن اي خطأ موجود في الكتاب .

الكتاب تحت رعاية شركه اوت بوكس

www.outbox-eg.com

لمتابعه مواضيعي في الـ Developer والـ ERP من خلال

Facebook

<http://www.facebook.com/pages/Oracle-Out-Box/167082563407931>

Youtube

<http://www.youtube.com/outboxahmedhabib>

حقوق الطبع والنشر والتوزيع محفوظة ولا يجوز لأي شخص نشره أو بيعه إلا بإذن كتابي مني

AHMED HABIB

Email: Dev_Habib@yahoo.com

Mobil: 002 0100 9844556

Oracle Developer
Instructor Oracle Developer
Oracle Financial Consultant
Instructor Oracle Financial R12

Index

<i>PL/SQL Fundamentals</i>	
<i>CH1 : Introduction to PL/SQL</i>	4
<i>CH2 : Declaring PL/SQL Variables</i>	9
<i>CH3 : Writing Executable Statements</i>	17
<i>CH4 : Interacting with the Oracle Server</i>	22
<i>CH5 : Writing Control Structures</i>	27
<i>CH6 : Working with Composite Data Types</i>	36
<i>CH7 : Using Explicit Cursors</i>	42
<i>CH8 : Handling Exceptions</i>	50
<i>PL/SQL Program Units</i>	
<i>CH1 : Creating Stored Procedures</i>	58
<i>CH2 : Creating Stored Functions</i>	67

PL/SQL Fundamentals

CHAPTER 1

Introduction to PL/SQL

PL/SQL

Procedural Language / Structure Query Language



ملحوظة :-


- سوف نقوم في الـ PL بعمل كود وتخزينه في الـ Database ونقوم بالنداء عليه فقط واستدعائه .
- سوف نستخدم ما يسمى بمتغيرات (Variables) وايضا ثوابت (Constant) وانواع اخري .
- سنستخدم جمل شرطيه (IF Statment) وايضا (Loop) وهي تكرار عمليه معينه عدد مرات معين .
- سوف نقوم بكتابه الكود مره واحده ولكننا نستطيع تنفيذه اكثر من مره .

مميزات الـ PL/SQL .

اي اننا نستطيع وضع اكثر من Block داخل بعضهما .	Modularized
يتكامل مع منتجات اوراكل مثل Jdeveloper	Integration
نستطيع الكتابه علي اي Operating System او Platform	Portability
نستطيع معالجه الاخطاء و اظهارها بالشكل الذي نريده	Exception

PL/SQL Block Structure

يتكون كود الـ PL من عدة Blocks وكل Block يتكون من :-

	Optional	وهي منطقة التعريف والاعلان وتحتوي علي Variables , Cursor , Constant , Other Types
	Mandatory	المنطقة التنفيذية SQL Statment , Pl/sql Statment
	Optional	معالجة الاخطاء
	Mandatory	نهاية الـ Block .

ملحوظة :-

يتكون الـ Block علي الاقل من Begin , End

Block Types

Anonymous

```
[DECLARE]

BEGIN
  --statements

[EXCEPTION]

END ;
```

Procedure

```
PROCEDURE name
IS
BEGIN
  --statements

[EXCEPTION]

END ;
```

Function

```
FUNCTION name
RETURN datatype
IS
BEGIN
  --statements
  RETURN value;

[EXCEPTION]

END ;
```

المقصود بالـ Subprogram وهو الـ Procedure و الـ Function .

Differences between Anonymous Blocks and Subprograms

<u>Anonymous Block</u>	<u>Subprogram</u>
ليس له اسم	لها اسم
نقوم بعمل Compile كل مره يُنفذ .	نقوم بعمل Compile له مره واحده فقط .
لا يُخزن في الـ Database	يُخزن في الـ Database
لا نستطيع تنفيذه من App اخري	نستطيع تنفيذه من App اخري
لا ياخذ Parameters	ممكن ان ياخذ Parameters

■ ماهو الفرق بين الـ Procedure و الـ Function ؟

- الـ Procedure ممكن تقود بقيمه او اكثر او لا تقود بقيمه .
- الـ Function لابد ان تعود بقيمه .

■ استخراج اسم الموظف رقم 100 ؟

```


ora41
0.50301397 seconds
Enter SQL Statement:
DECLARE
    v_fname VARCHAR2(20);
BEGIN
    SELECT first_name
    INTO v_fname
    FROM employees
    WHERE employee_id = 100;
END;

```

Anonymous Block Completed.

ملحوظة :-

- ◀ لابد من تهيئته البرنامج لاستخراج جمل الطباعة عن طريق كتابة Set Serveroutput On وتكتب مره واحده في الـ Session وتظل محفوظه حتي نقوم بإغلاقه .
- بعد تهيئته البرنامج نقوم بكتابه جملة الطباعة وهي DBMS_OUTPUT.PUT_LINE و PUT_LINE هي Procedure محفوظه داخل Package تُسمى DBMS_OUTPUT .
- ◀ نضع جملة الطباعة داخل Begin وتأخذ Parameters واحد لذلك لا نستطيع استخدام (,) داخلها بل نستخدم بدلا منها (||) في حاله اردنا طباعه اكثر من متغير .
- ◀ في حاله كتابه الكود بشكل سليم دون اخطاء فسوف تظهر . Anonimouse Block Completed.



OUT BOX

CHAPTER 2

Declaring PL/SQL Variables

Variables

وهي مكان في الميموري نقوم فيه بتخزين بيان معين او قيمه معينه تخزين مؤقت ونستطيع استخدام الـ Variables للتعديل او الاضافه علي القيم الموجوده في الـ Database وايضا نستطيع استخدامه اكثر من مره .

قواعد عامله لتسميه الـ Variables .

- ✓ لا بد ان يبدأ بحرف .
- ✓ من الممكن ان يحتوي علي حروف وارقام وعلامات .
- ✓ ألا يزيد عن 30 حرف .
- ✓ ألا يكون كلمه من كلمات Oracle المحجوزه مثل (Select) مثلا او غيرها .

ملحوظه :-

- ◀ نقوم بتعريف الـ Variable في الـ Declarative Section اي (Declare) .
- ◀ من الممكن اعطاء الـ Variable قيمه افتراضيه في الـ Declare ونستطيع تبديل هذه القيمه بقيمه اخري من خلال الـ Executable Section اي (Begin) .

Examples

DECLARE

<u>Variable Name</u>	<u>Datatype ;</u>
V_hire	Date;
V_deptno	Number Not Null := 10 ;
V_location	Varchar2(13) := 'Out Box' ;
V_comm	Constant Number := 1400 ;

Examples

I

DECLARE

V_Name Varchar2(20) ;

BEGIN

DBMS_OUTPUT.PUT_LINE ('My name is: '|| V_Name);

- في هذه الحالة لن يطبع شئ سوي كلمه My name is: فقط .

V_Name := 'Ahmed Habib';

DBMS_OUTPUT.PUT_LINE('My name is: '||V_Name);

- في هذه الحالة سيطبع كلمه My name is: Ahmed Habib

END ;

II

DECLARE

V_Name Varchar2(20) := 'Ahmed';

BEGIN

V_Name := 'Habib';

DBMS_OUTPUT.PUT_LINE('My name is: '||V_Name);

END ;

ملحوظه:-

في هذه الحالة ال Variable به قيمه في ال Declarative Section وايضا في

ال Executable Section ولكن الاقوي والاحق في التنفيذ هو ال

Executable ولذلك سيطبع ما بداخله وهو كلمه My name is: Habib .

Types of Variables

- **PL/SQL variables:**
 - Scalar.
 - Composite. (CH6)
 - Reference.
 - Large object (LOB).
- **Non-PL/SQL variables: Bind variables.**

ملحوظه :-

- ◀ من الافضل تسميه الـ Variable باسم يدل علي البيان الذي يحمله.
- ◀ من الافضل عدم تسميه الـ Variable علي اسم Column .
- ◀ الـ Variable الذي يكون به Constant او Not Null لابد من إعطائه قيمه افتراضيه في الـ Declare .
- ◀ نستطيع اعطاء الـ Variable قيمه افتراضيه عن طريق عمل Assignment Operator (:=) او كتابه كلمه Default مثل :-
- V_Name Varchar2(20) := 'Ahmed Habib' ;
- V_Name Varchar2(20) Default 'Ahmed Habib' ;
- ◀ من الافضل تعريف كل Variable علي سطر حتي يُسهل علينا عمليه القراءة وايضا معالجه الاخطاء وسهوله الوصول اليها .

- وسوف نتناول في هذا الـ Chapter النوع الاول وهو الـ Scalar وهو جزء من الـ Pl/sql Variables وايضا سنتناول النوع الاخير وهو الـ Non Pl/sql Variables .

Declaring Scalar Variables

DECLARE

```
V_Job          Varchar2 (9) ;
V_Count        Binary_Integer    := 0;
V_Dept         Number (9, 2)     := 0;
V_Orderdate    Date              := Sysdate+ 7;
V_Tax_Rate     Constant Number    := 8.25;
V_Valid        Boolean Not Null   := TRUE;
```

%TYPE Attribute

وهي Datatype وميزتها انها تأخذ نفس الـ Datatype الخاصه بالـ Column ونفس الـ Length .

DECLARE

```
Variable Name  Table Name . Column Name %Type ;
V_Name         Employees.Last_Name%Type    ;
V_Balance      Number(7,2) ;
V_Min_Bal      V_Balance%TYPE := 1000;
```

ملحوظه:-

- هذا النوع افضل من الـ Basic Scalar حيث انه عند عمل اي تعديل او تغيير في الجدول في الـ Length فان الـ Variable يتغير تلقائيا بتغيير الجدول .
- القيمه التي تعود بها Boolean هي True او False او Null .

■ استخرج اسم ومرتب الموظف الذي يحمل الرقم 100 ؟

DECLARE

V_Name Employees.Last_Name%Type ;

V_Sal Number ;

V_Id Employees.Employee_id%Type := 100 ;

BEGIN

Select Last_Name , Salary into V_Name , V_Sal

From Employees

Where Employee_Id = V_Id ;

DBMS_OUTPUT.PUT_LINE(V_Name || ' and His Salary is : ' || V_Sal);

END ;

Bind Variables

ملحوظة:-

- يتم انشائها داخل الـ Session وليس داخل الـ Block وتسمى Host Variable .
- يُوضع قبلها كلمة Variable وتستخدم ايضا في الـ Sql و الـ Pl/Sql.
- عند كتابه Bind Variable يكون شكله كالتالي

Variable Variable Name Datatype

- عند النداء عليه واستدعائه لابد ان نضع قبله Colon كالتالي Variable Name :
- نستطيع استخدام جملة الطباعة العادية مع الـ Bind Variable .

■ استخرج مرتب الموظف رقم 178 ؟

```
Variable V_Sal Number
BEGIN
    Select Salary into :V_Sal
    From Employees
    Where Employee_Id = 178 ;
END ;
/
Print V_Sal
```

ملحوظة :-

◀ يمكننا استخدام الـ Bind Variable في الـ SQL .

مثال

الان الـ Bind Variable المُسمي V_Sal المذكور في المثال السابق يحتوي علي مرتب الموظف رقم 178 ونفترض اننا نريد الاستعلام عن الموظفين الذين يحصلون علي هذا المرتب ؟

```
Select Lat_Name
From Employees
Where Salary = :V_Sal ;
```

◀ هناك جملة لا تعمل الا مع الـ Bind Variable وهي تستخدم في الطباعة

اتوماتيكيا معه فقط وتسمي Set Autoprint On .

Prompt for Substitution Variables

- قم باستخراج مرتب الموظف الذي ساعطيك رقمه ايا كان مستخدماً
الـ Bind Variable للمرتب ؟

Set Veriffy Off

Variable V_Sal Number

Accept Empno Prompt ' Please Enter a Valid Employee ID '

Set Autoprint On

DECLARE

V_Id Number(6) := & Empno ;

BEGIN

Select Salary into :V_Sal

From Employees

Where Employee_Id = V_Id ;

END ;

OUT BOX

CHAPTER 3

Writing Executable Statements

Lexical Units in a PL/SQL Block

ملحوظة:-
0

- ◀ عند التعامل مع الاحرف او التواريخ فلا بد من وضعها بين Single Quote .
- ◀ نستطيع كتابه الـ Code علي اكثر من سطر .
- ◀ نستطيع عمل Comment للـ Code في حالتين :-
- 1- لعمل Comment لسطر واحد نستخدم -- .
- 2- لعمل Comment لكثر من سطر نستخدم /* */ .
- ◀ من الافضل ان نقوم بكتابه Comment علي الاكواد لشرح وظيفته وحتى اذا جاء مبرمج اخر ليكمل المشروع او كان معك في نفس المشروع يستطيع ان يفهم وظيفته .

SQL Functions in PL/SQL

نستطيع استخدام جميع الـ Functions الموجوده في الـ SQL بداخل الـ PL/SQL
ماعدا Decode & Group Functions .

DECLARE

V_Lname Varchar2(20) := Initcap(' KING') ;

V_Name Varchar2(20) ;

BEGIN


Select First_Name into V_Name

From Employees

Where Last_Name Like V_Lname And Salary = 24000 ;

END;

Nested Blocks



ملحوظة:-

البلوك الداخلي يري البلوك الخارجي ويؤثر فيه
والعكس غير صحيح اي ان البلوك الخارجي لا
يري البلوك الداخلي .

Examples

```
DECLARE
  V_Father_Name  Varchar2(20) := 'Habib'      ;
  V_Date_of_Birth Date      := '20-Apr-1972'  ;
BEGIN
  DECLARE
    V_Child_Name  Varchar2(20) := 'Ahmed'      ;
    V_Date_of_Birth Date      := '12-Dec-2002' ;
  BEGIN
    DBMS_OUTPUT.PUT_LINE( V_Father_Name );
    DBMS_OUTPUT.PUT_LINE( V_Date_of_Birth );
    DBMS_OUTPUT.PUT_LINE( V_Child_Name );
  END;
  DBMS_OUTPUT.PUT_LINE( V_Date_of_Birth );
END;
```

ملحوظة:-

نستطيع عمل Qualifier للبلوك وتسميته باي اسم نريده حتي تتمكن من تحديد
البلوك المراد النداء عليه ولكن لابد من وضعه بين <> .
لو هناك متغير في البلوك الكبير وهو نفس الاسم في البلوك الصغير وقمنا
بالنداء عليه من داخل البلوك الصغير فالاولويه للمتغير الموجود في البلوك
الصغير .

Operators in PL/SQL

- Logical
- Arithmetic
- Concatenation
- Parentheses to control order of operations

Same in as
SQL

Qualify an Identifier

▪ Example (I)

<< Box >>

DECLARE

V_Father_Name Varchar2(20) := 'Habib' ;

V_Date_of_Birth Date := '20-Apr-1972' ;

BEGIN

DECLARE

V_Child_Name Varchar2(20) := 'Ahmed' ;

V_Date_of_Birth Date := '12-Dec-2002' ;

BEGIN

DBMS_OUTPUT.PUT_LINE(V_Father_Name); ➡ Habib

DBMS_OUTPUT.PUT_LINE(Box.V_Date_of_Birth); ➡ 20-Apr-1972

DBMS_OUTPUT.PUT_LINE(V_Child_Name); ➡ Ahmed

DBMS_OUTPUT.PUT_LINE(V_Date_of_Birth); ➡ 12-Dec-2002

END;

END;

▪ Example (II)

<< Outbox >>

DECLARE

V_Sal Number(7,2) := 60000 ;

V_Comm Number(7,2) := V_Sal * 0.20;

V_Message Varchar2(255) := ' eligible for commission' ;

BEGIN

DECLARE

V_Sal Number(7,2) := 50000 ;

V_Comm Number(7,2) := 0 ;

V_Total Number(7,2) := V_Sal + V_Comm ;

BEGIN

V_Message := 'CLERK not' || V_Message ;

DBMS_OUTPUT.PUT_LINE (V_Message);

Outbox.V_Comm := V_Sal * 0.30;

DBMS_OUTPUT.PUT_LINE (Outbox.V_Comm);

END;

V_Message := 'SALESMAN' || v_message;

DBMS_OUTPUT.PUT_LINE (V_Message);

END;

- | | | |
|----------|---|------------------------------------|
| 1 | ➔ | eligible for commission . |
| 2 | ➔ | 15000 |
| 3 | ➔ | SALESMAN eligible for commission . |

OUT BOX

CHAPTER 4

Interacting with the Oracle Server

Using PL/SQL to Manipulate Data

INSERT	UPDATE	DELETE	MERGE
--------	--------	--------	-------

Inserting Data

```
BEGIN
  INSERT into Departments
  Values (Dept_Seq.nNextval , 'Habib_Dept', 100 , 1700);
END;
```

Updating Data

```
DECLARE
  Sal_Increase Employees.Salary%Type := 800 ;
BEGIN
  UPDATE Employees
  Set Salary = Salary + Sal_Increase
  Where Job_id = 'ST_CLERK';
END;
```

Deleting Data

DECLARE

Deptno Employees.Department_id%Type: = 10;

BEGIN

DELETE From Employees

Where Department_id = Deptno;

END;

SQL Cursor

<u>IMPLICIT</u>	<u>EXPLICIT</u>
وهو من النوع الضمني اي ان Oracle هي التي قامت بعمله ويعمل تلقائيا بمجرد النداء عليه .	والمبرمج هو الذي يقوم بعمله وسوف نتناوله بالشرح في (7) Chapter .

ملحوظة :-

Cursor يعتبر مساحة في الميموري تقوم اوراكل بتجهيزها اتوماتيكيا مع كل جملة SQL وتضع بها البيانات القادمة من الـ SQL .

SQL Cursor Attributes for Implicit Cursors

SQL%FOUND	لو وجد بيانات يقوم باسترجاع True ولو لم يجد بيانات يسترجع False .
SQL%NOTFOUND	لو لم يجد بيانات يقوم باسترجاع True وان وجد بيانات يسترجع False .
SQL%ROWCOUNT	يقوم باسترجاع عدد الصفوف التي تأثرت بالعملية في الميموري.

- قم بمسح الموظف رقم 176 ومن ثم اطبع عدد الصفوف التي تأثرت بالعملية ان وُجد هذا الموظف ؟

DECLARE

V_Rows_Deleted Varchar2(30) ;

V_Empno Employees.Employee_id%Type := 176;

BEGIN

Delete From Employees

Where Employee_id = V_Empno ;

V_Rows_Deleted := (SQL%Rowcount || ' row deleted.');

DBMS_OUTPUT.PUT_LINE (V_Rows_Deleted);

END;

- قم باعطاء الموظف رقم 100 ان وُجد المرتب الاتي 20000 .

DECLARE

V_Name Varchar2 (20);

BEGIN

Select Last_Name into V_Name

From Employees

Where Employee_Id = 100;

IF SQL%Found Then

Update Employees

Set Salary = 20000

Where Employee_Id = 100;

End IF;

END;


ملحوظه :-

◀ ولقد قمنا بعمل جملة الـ Select حتي نعرف هل يوجد موظف بهذا الرقم ام لا
فلو وجد سيأتي باسمه وان لم يجد فلن يأتي بشي .
◀ اما SQL%Found فانها تنتظر الي الـ SQL هل وجدت بيانات فان وجدت
فتقوم بتنفيذ عمليه التعديل وان لم تجد فلن تفعل شئ .

OUT BOX

CHAPTER 5

Writing Control Structures



IF Statement

```

DECLARE
  V_Myage Number;
BEGIN
  IF V_Myage < 11 Then
    DBMS_OUTPUT.PUT_LINE (' I am a child ');
  ELSE
    DBMS_OUTPUT.PUT_LINE (' I am not a child ');
  End IF;
END;
```

Example

```

DECLARE
  V_Myage Number:=31;
BEGIN
  IF      V_Myage < 11 Then
    DBMS_OUTPUT.PUT_LINE (' I am a child ');
  ELSIF V_Myage < 20 Then
    DBMS_OUTPUT.PUT_LINE (' I am young ');
  ELSIF V_Myage < 30 Then
    DBMS_OUTPUT.PUT_LINE (' I am in my twenties');
  ELSIF V_Myage < 40 Then
    DBMS_OUTPUT.PUT_LINE (' I am in my thirties');
  ELSE
    DBMS_OUTPUT.PUT_LINE (' I am always young ');
  End IF;
END;
```

ملحوظه :-

◀ لو الـ Condition ← Null فنقوم بتنفيذ ما داخل Else مباشرة .

◀ نستطيع وضع اكثر من جملة IF داخل بعضها .

◀ كلمة Else تأتي مره واحده في النهايه .

◀ IF Condition Then Action

True → سوف ينفذ الـ Action

False → فلن ينفذ الـ Action

◀ لابد من وضع ; بعد كل IF .

CASE Expressions

DECLARE

V_Grade Char(1) := Upper('&Grade');

Appraisal Varchar2 (20);

BEGIN

Appraisal := CASE

When V_Grade = 'A' Then 'Excellent'

When V_Grade In ('B','C') Then 'Good'

Else 'No such grade'

End;

DBMS_OUTPUT.PUT_LINE ('Grade: ' || V_Grade || ' Appraisal ' || Appraisal);

END;

ملحوظة:-

- ◀ من الممكن ان يأتي بعد Then جملة Select .
- ◀ Searched Case وهو ان يأتي بعد When أكثر من Exp .

Handling Nulls

ملحوظة:-

- ◀ المقارنات البسيطة لو فيها Null الناتج يصبح Null .
- ◀ لو وضعنا Not مع Null الناتج يصبح Null .
- ◀ And الاقوي هو الـ False والاضعف هو الـ True .
- ◀ Or الاقوي هو الـ True والاضعف هو الـ False .

LOOP Statements

وهو تكرار عملية معينة عدد مرات معين .

Basic Loop	While Loop	For Loop
------------	------------	----------

Basic Loop

Syntax

```
Loop
Statement;
Exit [When Condition];
End Loop;
```

■ قم بإضافه 3 مواقع باستخدام الـ Basic Loop .

DECLARE

V_Country_id Locations.Country_id%Type:= 'CA';

V_Loc_id Locations.Location_id%Type;

V_Counter Number := 1;

V_New_City Locations.City%Type := 'Montreal';

BEGIN

Select Max(Location_id) Into V_Loc_id From Locations

Where Country_id = V_Country_id;

LOOP

Insert Into Locations (Location_id, City, Country_id)

Values ((V_Loc_id + V_Counter), V_New_City, V_Country_id);

V_Counter := V_Counter + 1;

EXIT WHEN V_Counter > 3;

END LOOP;

END;

ملحوظة:-

◀ هل ستظل الـ Loop تعمل الي مالا نهايه ؟

بالطبع لا , فلابد ان نقوم بتعريف وعمل متغير يكون هو بمثابة العداد الذي يمثل عدد الدورات التي تقوم بها الـ Loop وفي كل دوره نضيف واحد علي هذا العداد حتي يصل الي الحد المطلوب ويخرج . اذن فالـ Basic Loop لابد من عمل متغير له .

◀ اذا لم نحدد Exit When Condition فسوف تظل الـ Loop تعمل وسوف تتوقف الـ Database عن العمل بسبب التهنيج او يحتوي يظهر هذا الايرون Numeric or Value Error :Number Precision too Large .

◀ فرضا قمنا بعمل $Exit\ When\ Counter < 1$ فإن الـ Basic Loop تقوم بالتنفيذ مره واحده رغم ان الشرط يخالف الـ Loop ولكنها هكذا سوف تنفذ مره واحده علي الاقل لان الـ Statement تُنفذ اولا قبل المرور علي الـ Condition .

While Loop

Syntax

```
While Condition Loop
Statement;
End Loop;
```

■ قم بإضافه 3 مواقع باستخدام الـ While Loop .

DECLARE

V_Country_id Locations.Country_id%Type:= 'CA';

V_Loc_id Locations.Location_id%Type;

V_Counter Number := 1;

V_New_City Locations.City%Type := 'Montreal';

BEGIN

Select Max(Location_id) Into V_Loc_id From Locations

Where Country_id = V_Country_id;

WHILE v_counter <= 3 LOOP

Insert Into Locations (Location_id, City, Country_id)

Values ((V_Loc_id + V_Counter), V_New_City, V_Country_id);

V_Counter := V_Counter + 1;

END LOOP;

END;

ملحوظه:-

◀ طالما الـ Condition يتحقق تظل الـ Loop تعمل .

◀ الـ While Loop تقوم بعم Check علي الـ Condition اولا قبل تنفيذ

الـ Statement وذلك عكس الـ Basic Loop . ولا بد من عمل متغير لها

كعداد .

For Loop

وهو النوع الأفضل والأكثر والاسهل استخداما.

Syntax

```
For Counter in [Reverse] lower-Bound .. Upper-Bound Loop
    Statement;
End Loop;
```

ملحوظة:-

◀ لا تحتاج الي عمل عداد لها بل تقوم اوراكل بعمله اتوماتيكيا ولكننا نقوم بتسميته فقط .

◀ لو مثلا اردنا طباعه الارقام من 1 وحتى الرقم 10 فاننا نقوم بكتابه الجملة

بهذا الشكل For I in 1 .. 10 Loop DBMS_OUTPUT.PUT_LINE(I);

ولو اردنا طباعه الارقام عكسيا اي من الرقم 10 الي الرقم 1 مثلا فنكتبها

بهذا الشكل For I in 1 .. 10 Reverse Loop DBMS_OUTPUT.PUT_LINE (I);

◀ لا نستطيع استخدام Null في بدايه العد و اقل قيمه نبدأ بها هي 1 .

◀ نستطيع استخدام اكثر من Loop داخل بعضهما .

■ قم بإضافه 3 مواقع باستخدام الـ While Loop .

```
DECLARE
    V_Country_id  Locations.Country_id%Type:= 'CA';
    V_Loc_id      Locations.Location_id%Type;
    V_New_City    Locations.City%Type := 'Montreal';
BEGIN
    Select Max(Location_id) Into V_Loc_id From Locations
    Where Country_id = V_Country_id;
    FOR i in 1..3 LOOP
        Insert Into Locations (Location_id, City, Country_id)
        Values ((V_Loc_id + i), V_New_City, V_Country_id);
    END LOOP;
END;
```

OUT BOX

CHAPTER 6

Working with Composite Data Types

Composite Data Types

■ وهناك نوعان وهما :-

- PL/SQL Records
- PL/SQL Collections:-
 - ✓ Index By tables.
 - ✓ Nested table.
 - ✓ Varray.

PL/SQL Records

Records

ملحوظة:-

- الـ Record يحمل داخله اكثر من قيمة علي عكس الـ Scalar .
- يحتوي الـ Record علي Fields وتتم عملها بواسطة مايسمي (3GL) اي 3 Generation Language ويتم انشاء هذه الخانات بنفس الطريقة التي تستخدمها هذه اللغات C و C++ .
- المبرمج يقوم بتعريف الـ Datatype الخاصه بهذه الـ Fields وتسميتها ايضا ثم عمل الـ Variable واعطاء الـ Record له .
- دعونا نتفق ان اي متغير لا بد له من نوع بيانات اي ان اي Variable له Datatype. ولكن الاختلاف هنا ان الداتا تيب هذه المره عباره عن مجموعه من الداتا تيب وليست واحده ولذلك يجب علينا انشاء الداتا تيب اولا وبعد ذلك اعطائها للمتغير .

Syntax Create Datatype and Variable .

Type Type_Name is Record (Field1 Datatype , Field2 Datatype ,) ;
Variable ↓ ;

ملحوظة:-

◀ نري اننا قد قمنا بعمل داتا طيب وتحتوي علي 2 Fields وكاننا قمنا بعمل متغيران داخل الداتا تيب هذه وقمنا باعطاء الداتا تيب هذه اسما هذا الاسم قمنا بعطائه للمتغير حتي يتمكن من ان يحمل داخله اكثر من قيمه.

Field1 (data type)	Field2 (data type)	Field3 (data type)
Employee_id Number(6)	Last_Name Varchar2(25)	Job_Id Varchar2(10)
100	King	AD_PRES

▪ استخراج اسم ووظيفه الموظف رقم 100 باستخدام Record .

```

DECLARE
Type    Emp_Rec is Record (R_Name Varchar2(20) , R_Job Varchar2(20));
V_Rec   Emp_Rec ;
BEGIN
Select Last_Name , Job_id into V_Rec
From Employees
Where Employee_id = 100 ;
DBMS_OUTPUT.PUT_LINE (V_Rec.R_Name || ' ' || V_Rec.R_Job ) ;
END;
```

ملحوظة:-

◀ في حاله الطباعه لابد من كتابه اسم المتغير ثم اسم الفيلد .
◀ عند الاستعلام ووضع القيم داخل المتغير يجب ان يراعي الترتيب الموجود في الـ Record .

%ROWTYPE Attribute

Syntax

Variable_Name Table_Name%Rowtype ;

ملحوظة:-

➤ ونستخدم %Rowtype ان اردنا استخراج جميع اعمده الجدول لاننا ان استخدمنا الطريقة العادية وهي الـ Scalar فاننا سنحتاج الي تعريف متغيرات بعدد اعمده الجدول وكذلك لو قمنا بعمل Record فاننا سنحتاج الي عمل Fields بعدد اعمده الجدول ايضا .

▪ استخراج جميع بيانات الموظف رقم 100 . وقم بطباعه اسمه ومرتبته ؟

```
DECLARE
  V_Rec Employees%Rowtype ;
BEGIN
  Select * into V_Rec
  From Employees
  Where Employee_Id = 100 ;
  DBMS_OUTPUT.PUT_LINE ( V_Rec.Last_Name || ' ' || V_Rec.Salary);
END;
```

مثال اخر

```
DECLARE
  Type T_Rec is Record (R_Sal Number, R_Minsal Number Default 1000,
                        R_Hire_Date Employees.Hire_Date% Type,
                        R_Rec1 Employees%Rowtype );
  V_Myrec T_Rec ;
BEGIN
  V_Myrec.R_Sal      := V_Myrec.R_Minsal + 500;
  V_Myrec.R_Hire_Date := S ysdate;
  Select * into V_Myrec.R_Rec1
  From Employees
  Where Employee_Id = 100;
  DBMS_OUTPUT.PUT_LINE (V_Myrec.R_rec1.Last_Name || ' ' ||
                        To_Char(V_Myrec.R_Hire_Date) || ' ' ||
                        To_Char(V_Myrec.R_Sal) );
END;
```

PL/SQL collections

INDEX BY Tables

وهي نوع من انواع الداتا تيب التي نقوم بعملها وهو يتكون من عمودين العمود الاول يكون Primary Key ويفضل ان تكون نوع البيانات به رقميه والعمود الثاني اما ان يكون Scalar او Composite .

Unique key

...
1
5
3
...

Number

Value

...
Jones
Smith
Maduro
...

Scalar

Syntax

TYPE Type_Name Is Table Of
(Column_type | Table.Column%Type | Table%Rowtype)
INDEX BY (Datatype);

Variable_Name Type_Name;

ملحوظه:-

Exists	تعود ب True لو وجدت قيمه في الـ Index .
Count	بتقوم بعد الـ Rows الموجوده في الـ Index .
Next	نحدد قيمه وهي تأتي بالقيمه التاليه لها في الـ Index .
Prior	نحدد قيمه وهي تأتي بالقيمه السابقه لها في الـ Index .
Delete	تقوم بمسح القيم الموجوده في الـ Index .
Trim	نحدد لها قيمه وتمسح ما بعدها فقط .
First&Last	تأتي بأول قيمه واخر قيمه في الـ Index .

Examples

I

```
DECLARE
  Type T_Name_Table Is Table Of  Employees.Last_name%Type
  Index By Number;
  V_Name      T_Name_Table;
BEGIN
  V_Name (1)  := 'Ahmed Habib';
  IF V_Name.Exists(1) Then
    Insert into Employees (LAST_Name , Employee_Id , Hire_Date , Job_Id , Email )
    Values(V_Name(1) , 800 , '1-JAN-2010' , 'SA_REP' , 'Dev_Habib@Yahoo.com' );
  END;
```

II

```
DECLARE
  TYPE Emp_Table_Type Is Table Of  Employees%Rowtype
  INDEX BY PLS_INTEGER;
  V_Emp      Emp_Table_Type;
  V_Max_Count      Number := 104;
BEGIN
  FOR i IN 100 .. V_Max_Count  LOOP
    Select * Into V_Emp(i)
    From Employees
    Where Employee_Id = i;
  END LOOP;
  FOR i IN V_Emp.FIRST .. V_Emp.LAST  LOOP
    DBMS_OUTPUT.PUT_LINE ( V_Emp(i).Last_Name);
  END LOOP;
END;
```


OUT BOX

CHAPTER 7

Using Explicit Cursors

■ ماهو ال Cursor وما هي فائدته ؟

وهو ليس من النوع الضمني اي ان المبرمج هو الذي يقوم بعمله وفتحته وتعبئته وغلقه. وهو عبارة عن مكان في اليمموري نقوم فيه بعمل عملياته اي اننا نقوم بتخزين كل البيانات التي نحتاجها من الداتا بيز في ال Cursor عوضا ان نذهب في كل مره الي الداتا بيز وهذا بالطبع اسرع كثيرا ويوفر وقتا كبير اذن ففائدته الكبرى هي السرعة في الاستعلام عن البيانات فبدل من ان نضع القيمة من الداتا بيز في المتغير وهو ايضا مكانه في اليمموري ان ناتي بكل البيانات في اليمموري مره واحده ومن ثم نقلها الي المتغير .



Syntax

```

DECLARE
  Cursor Cursor_Name is Select_Statement;
BEGIN
  Open Cursor_Name ;
  Fetch Cursor_Name Into Variables ;
  Close Cursor_Name ;
END;
```

■ استخراج ارقام واسماء موظفي الاداره 30 ؟

```

DECLARE
    Cursor Emp_Cursor is  Select Employee_Id , Last_Name
                          From Employees
                          Where Department_Id = 30 ;

    V_Id      Employees.Employee_Id%Type ;
    V_Name    Employees.Last_Name%Type ;

BEGIN
    Open Emp_Cursor ;
    LOOP
        Fetch Emp_Cursor into V_Id , V_Name ;
        Exit When Emp_Cursor %Notfound ;
        DBMS_OUTPUT.PUT_LINE ( V_Id || ' ' || V_Name );
    End Loop ;
    Close Emp_Cursor ;
END;

```

■ قم بحل المثال السابق ولكن باستخدام متغيرات من النوع Composite وبالتحديد باستخدام الـ Record بدلا من المتغيرات الـ Scalare ؟

```

DECLARE
    Cursor Emp_Cursor is  Select Employee_Id , Last_Name
                          From Employees
                          Where Department_Id = 30 ;

    V_Rec      Emp_Cursor %Rowtype ;

BEGIN
    Open Emp_Cursor ;
    LOOP
        Fetch Emp_Cursor into V_Rec;
        Exit When Emp_Cursor %Notfound ;
        DBMS_OUTPUT.PUT_LINE ( V_Rec.Employee_Id || ' ' ||
                               V_Rec.Last_Name );
    End Loop ;
    Close Emp_Cursor ;
END;

```

ملحوظه :-

➤ لقد قمنا بعمل الـ Cursor ووضعنا به كل البيانات التي نريديها من الداتا بيز ومن ثم بدانا بسحبها من الـ Cursor الي المتغير وبسرعه لان الاثنان في الميموري سواء المتغير او الـ Cursor .

Explicit Cursor Attributes

<u>Attribute</u>	<u>Type</u>	<u>Description</u>
%ISOPEN	Boolean	تعود بقيمه True لو مفتوح والعكس.
%NOTFOUND	Boolean	تعود بقيمه True لو ال Cursor فاضي والعكس.
%FOUND	Boolean	تعود بقيمه True لو ال Cursor به بيانات والعكس.
%ROWCOUNT	Number	تعود بعدد الصفوف التي تم سحبها من ال Cursor.

Cursor FOR Loops

استخدام ال For Loop مع ال Cursor يمثل شيئا رائعا حيث اننا لا نحتاج الي تعريف متغير بالنسبه لل For وايضا لا نحتاج لفتح ولا تعبئه ولا غلق ال Cursor في حاله استخدامنا ال For مع ال Cursor. اذن فلا احتاج الا لعمل ال Cursor فقط وتقوم اوراكل اتوماتيكيا بفتحه وتعبئته وغلقه .

■ استخراج ارقام واسماء موظفي الاداره 30 باستخدام ال For Loop ؟

```

DECLARE
    Cursor Emp_Cursor is  Select Employee_Id , Last_Name
                           From Employees
                           Where Department_Id = 30 ;

BEGIN
    For Rec in Emp_Cursor Loop
        DBMS_OUTPUT.PUT_LINE ( Rec.Employee_Id || ' ' ||
                               Rec.Last_Name );
    End Loop ;
END;
```

Syntax %ISOPEN Attribute

```
BEGIN
  IF NOT c_emp_cursor%ISOPEN THEN
    OPEN c_emp_cursor;
  END IF;
  LOOP.....
```

▪ مثال علي استخدام %ROWCOUNT and %NOTFOUND .

```
DECLARE
  Cursor C_Emp Is Select Employee_Id, last_name
                  From Employees;
  V_Rec C_Emp%Rowtype;
BEGIN
  Open C_Emp ;
  LOOP
    Fetch C_Emp INTO v_emp_record;
    Exit When C_Emp %Rowcount > 10 or C_Emp %Notfound;
    DBMS_OUTPUT.PUT_LINE ( V_Rec.Employee_Id || ' ' ||
                          V_Rec.Last_Name);

  End Loop;
  Close C_Emp ;
END ;
```

ملحوظه:-

➤ في هذه الحاله اي شرط يتحقق اولا تخرج الـ Loop سواء الـ Cursor اصبح فارغا او عدد الصفوف التي تم سحبها اكبر من 10.

▪ مثال علي استخدام الـ Cursor FOR Loops Using Subqueries .

```
BEGIN
  For R in ( Select Employee_Id, Last_Name
            From Employees
            Where Department_Id =30 ) LOOP

    DBMS_OUTPUT.PUT_LINE ( R.Employee_Id || ' ' || R.Last_Name);
  End Loop;
END;
```

ملحوظه:-

➤ كل ما فعلناه اننا بدلا من عمل الـ Cursor ووضع الجمله داخله اننا وضعناها مكان الـ Cursor في الـ For Loop فقط وتقوم بنفس الوظيفه ولكن ليس بنفس السرعه .

▪ مثال علي استخدام الـ Cursors with Subqueries .

DECLARE

Cursor Sub is Select D.Department_Id , D.Department_Name,
E.Staff
From Departments D, (Select Department_Id,
Count (*) as Staff
From Employees
Groub by Department_Id) E
Where D.Department_Id = E.Department_Id
And E.Staff >= 3;

BEGIN

ملحوظه:-

➤ وناتج هذا الـ Cursor هو ارقام واسماء الادارات التي بها عدد موظفين اكبر
من او يساوي 3 موظفين .

Cursors with Parameters

وفائدتها اننا نستطيع تغيير القيم الناتجة عن الـ Cursor عن طريق ادخال Parameters لتغيير النواتج .

Example

```

DECLARE
Cursor C_Emp (P_Id Number) is Select Employee_Id, Last_Name
                                From Employees
                                Where Department_Id = P_Id ;

V_Id      Number ;
V_Name    Varchar2(20) ;

BEGIN
Open C_Emp( 10 ) ;
Loop
    Fetch C_Emp Into V_Id , V_Name ;
Exit When C_Emp%Notfound ;
DBMS_OUTPUT.PUT_LINE (V_Id || ' ' || V_Name);
End Loop ;
Close C_Emp ;

DBMS_OUTPUT.PUT_LINE ('-----');

Open C_Emp( 30 ) ;
Loop
    Fetch C_Emp Into V_Id , V_Name ;
Exit When C_Emp%Notfound ;
DBMS_OUTPUT.PUT_LINE (V_Id || ' ' || V_Name);
End Loop ;
Close C_Emp ;

END;
```

ملحوظة:-

➤ بهذا الشكل نكون قد قمنا بتغيير البيانات اكثر من مره عن طريق فتح الـ Cursor مره اخري ببيانات اخري ولكن ذلك يتطلب منا غلق الـ Cursor القديم والا سيظهر لنا Error انه مفتوح بالفعل . ولتغلب علي هذه المشكله نستخدم Cursor_Name%Isopen .

OUT BOX

CHAPTER 8

Handling Exceptions

وكما ذكرنا من قبل ان هذه المرحلة هي مرحلة اختياريه نستطيع الاستغناء عنها ولكنها مرحلة هامه وفيها يتم التعامل مع الاخطاء التي من الممكن ان تظهر للمستخدم النهائي ومعالجتها واظهارها له بصوره يستطيع فهمها وقرائتها حيث ان الاخطاء التي تظهر من اوراق لا يستطيع المستخدم النهائي فهمها ولذلك نظهرها له بصوره بسيطه .

Exception Types

Predefined Oracle Server	Implicitly Raised	متعارف عليه وضمني
Non-predefined Oracle Server	Implicitly Raised	غير متعارف عليه وضمني
User-defined	Explicitly Raised	المبرمج هو الذي يقوم بعمله والتحكم به

Predefined Oracle Server

وهناك حوالي 21 Error متعارف عليهم من قبل اوراق وهم المشهورين ويكون لديهم اسم لل Error وايضا رقم ونص رساله .

Example

```

DECLARE
  V_Name Varchar2(20);
BEGIN
  Select First_Name into V_Name
  From Employees
  Where Last_Name Like 'King' ;
  DBMS_OUTPUT.PUT_LINE ('V_Name');
EXCEPTION
  When Too_Many_Rows Then
    DBMS_OUTPUT.PUT_LINE ('Query Retrieved Multiple Rows');
END;
```

ملحوظه:-

➤ Syntax Exception

EXCEPTION

```

When Exception1 [Or Exception2] Then Statement ;
[When ..... Then Statement ; ]
[When Others Then Statement ; ]
```

➤ لو وجد اكثر من موظف يُسمى King فانه سوف يظهر Error انه هناك اكثر من شخص ولكن مع كتابه الـ Exception فستظهر الرساله التي كتبناها وهذا الـ Error من المشاهير اي من الـ 21 .

➤ لو قمنا بكتابه هذه الجملة :-

Select Last_Name From employees Where salary = 2542184518 ;
فانه سوف يظهر لنا هذا الايرور No Data Found لانه لا يوجد موظف ياخذ هذا المرتب .

```
DECLARE
  V_Name Varchar2(20);
  V_Sal Number;
BEGIN
  Select Last_Name into V_Name
  From employees
  Where salary = 2542184518;
  DBMS_OUTPUT.PUT_LINE ( 'V_Name' );
  DBMS_OUTPUT.PUT_LINE ( 'V_Sal' );
EXCEPTION
  When No_Data_Found Then
    DBMS_OUTPUT.PUT_LINE ( 'لا يوجد موظف يأخذ هذا المرتب' );
  When Others Then
    DBMS_OUTPUT.PUT_LINE (SQLCODE || ' ' || SQLERRM);
END;
```

ملحوظه:-

➤ بهذه الصوره سوف يظهر انه لا يوجد موظف بهذا المرتب ولن يقوم بطباعه الـ V_Sal ولا الـ V_Name لانهما جاءا بعد جملة الـ Select التي تحتوي علي الايرور اما لو وضعنا جملة طباعه المرتب والاسم اعلي الـ Select فانها سوف تقوم بطباعه Null .

SQLCODE	تأتي برقم الـ Error .
SQLERRM	تأتي بنص رساله الـ Error .

▪ **ونقوم الان بعرض الـ Error المتعارف عليها والمشهوره وكما قلنا من قبل انها حوالى 21 Error وهم كالاتى :-**

Oracle Exception Name	Oracle Error	Explanation
DUP_VAL_ON_INDEX	ORA-00001	You attempted to create a duplicate value in a field restricted by a unique index.
TIMEOUT_ON_RESOURCE	ORA-00051	A resource timed out, took too long.
TRANSACTION_BACKED_OUT	ORA-00061	The remote portion of a transaction has rolled back.
INVALID_CURSOR	ORA-01001	The cursor does not yet exist. The cursor must be OPENed before any FETCH cursor or CLOSE cursor operation.
NOT_LOGGED_ON	ORA-01012	You are not logged on.
LOGIN_DENIED	ORA-01017	Invalid username/password.
NO_DATA_FOUND	ORA-01403	No data was returned
TOO_MANY_ROWS	ORA-01422	You tried to execute a SELECT INTO statement and more than one row was returned.
ZERO_DIVIDE	ORA-01476	Divide by zero error.
INVALID_NUMBER	ORA-01722	Converting a string to a number was unsuccessful.
STORAGE_ERROR	ORA-06500	Out of memory.
PROGRAM_ERROR	ORA-06501	Generic "Contact Oracle support" message.
VALUE_ERROR	ORA-06502	You tried to perform an operation and there was a error on a conversion, truncation, or invalid constraining of numeric or character data.
ROWTYPE_MISMATCH	ORA-06504	
CURSOR_ALREADY_OPEN	ORA-06511	The cursor is already open.
ACCESS_INTO_NULL	ORA-06530	
COLLECTION_IS_NULL	ORA-06531	
SELF_IS_NULL		Your program attempts to call a MEMBER method on a null instance. That is, the built-in parameter SELF (which is always the first parameter passed to a MEMBER method) is null.
SUBSCRIPT_BEYOND_COUNT		Your program references a nested table or varray element using an index number larger than the number of elements in the collection
SUBSCRIPT_OUTSIDE_LIMIT		Your program references a nested table or varray element using an index number (-1 for example) that is outside the legal range.
SYS_INVALID_ROWID		The conversion of a character string into a universal rowid fails because the character string does not represent a valid rowid.

Non-predefined Oracle Server

في هذه الحالة الـ Error لا يمتلك هوية اي لا يمتلك اسم ولديه بالطبع رقم ولديه ايضا رساله ولكن المشكله تكمن في اننا لا نستطيع النداء عليه لانه ليس له اسم ولذلك فنقوم بعمل متغير من النوع Exception اي انه يحمل بداخله Error ونضع الايرونر بداخله عن طريق ما يسمى Pragma وهي التي تربط وتضع الايرونر داخل الـ Variable .

- ما الذي سيحدث اذا حاولنا اصفه قيمه Null داخل عمود Not Null ؟
بالطبع سيظهر ايرونر وهو رقمه 01400- ورسالته تقول اننا لا نستطيع وضع قيم Null داخل عمود Nott Null ولكن المستخدم النهائي لو ظهرت له هذه الرساله وهذا الرقم لن يفهم شيئا ولكننا نريد ان نظهر له رساله يستطيع فهمها .

Example

DECLARE

V_Error Exception;
Pragma Exception_init (V_Error , -01400);

BEGIN

Insert into Departments (Department_Id , Department_Name)
Values (Null , 'OutBox_Habib');

EXCEPTION

When V_Error Then
DBMS_OUTPUT.PUT_LINE ('لا بد من وضع قيمه داخل هذا العمود');

END;

User-Defined Exceptions

وهما نوعان :-

Raise

Raise_Application_Error

✓ Raise

هذه المرة لا يوجد مشكله لانه يظهر لنا PL/SQL Procedural Successfully ولكن الكود لم يطبق فعليا ولم ينجح في تنفيذ المطلوب منه ولكنه بدلا من ان يظهر لنا ايرور يوضح لنا ان العملية لم تتم الا انه يظهر انها نجحت فعن طريق استخدام Raise نستطيع اذا لم ينفذ الكود المطلوب منه فاننا نجبره علي اظهار رساله توضع ان العملية لم تتم .

Example

```
DECLARE
  V_Error Exception ;
BEGIN
  Update Employees
  Set Salary = 12000
  Where Employee_Id = 123456 ;
  IF SQL%Notfound Then  Raise  V_Error ;
  End IF;
  DBMS_OUTPUT.PUT_LINE (SQL%Rowcount);
  Commit;
EXCEPTION
  When V_Error Then
    DBMS_OUTPUT.PUT_LINE ('لا يوجد موظف بهذا الرقم');
END;
```

ملحوظة:-

◀ لو لم نستخدم Raise لكان ظهر لنا انه تمت عملية التحديث بنجاح وهذا طبعا لم يحدث لعدم وجود موظف بهذا الرقم .

◀ **%ROWCOUNT** تقوم بالقراءة من خلال الميموري اي عدد الصفوف التي تأثرت بالعملية فاذا كنا كتبنا Commit قبل DBMS فاننا لن يظهر ولن يطبع شيئا لان Commit تقوم بعمل Save للعمليات من الميموري الي الداتا بيز وتمحي ما في الميموري.

✓ Raise Application Error

في هذه المره هناك ايرور او لا، لا يهم ولكن المبرمج يريد ان يظهر ايرور صريح مثل الايرور الخاص باوراكل اي برقم ورساله ولكن من الممكن ان نستخدم ارقام خاصه باوراكل ولكي نتجنب هذه المشكله قامت اوراكل بوضع مجموعه من الارقام تستطيع ان تستخدمها وهي تبدأ من 20000- وحتى 20999 - ونستخدم Raise_Application_Error حتي نقوم بهذا. ونستطيع ان نفعل ذلك مكانين وهما اما منطقه ال Exception Section او منطقه ال Executable Section.

• Executable Section

```
BEGIN
  Delete From Employees
  Where Employee_Id = 123456;
  IF SQL%Found Then
    DBMS_OUTPUT.PUT_LINE (SQL%Rowcount );
  Elsif SQL%Notfound Then
    Raise_Application_Error (-20001, 'لا يوجد موظف بهذا الرقم');

  End IF;
END;
```

• Exception Section

```
DECLARE
  V_Name Varchar2(20);
BEGIN
  Select Last_Name into V_Name
  From Employees
  Where Employee_Id = 123456;
EXCEPTION
  When No_Data_Found Then
    Raise_Application_Error (-20002, 'لا يوجد موظف بهذا الرقم');
END;
```

OUT BOX

PLSQL Program Units

CHAPTER 1

Creating Stored Procedures

وهو نوع من انواع البلوكات ويعتبر احد افراد عائلته الـ Subprogram وتكوينه شبيه بالبلوك العادي الـ Anonymous مع اختلاف بسيط ؛ مع الـ Procedure لا نستخدم الـ Declare ونستخدم بدلا منها Is او As ولا بد من كتابته Is او As حتي لو لم نضطر الي تعريف Variables اما والباقي كما هو . وميزه الـ Procedure اننا نقوم بكتابته الكود مره واحده ونستطيع تنفيذه والنداء عليه اكثر من مره . لانه مُخزن في الداتا بيز ولكنه ليس من الضروري ان يعود هذا الاجراء بقيمه معينه اي اننه ممكن ان يعود بقيمه او لا .

ما الفرق بين الـ Subprogram (Function , Procedure) و الـ Anonymous Block

<u>Anonymous</u>	<u>Subprogram</u>
ليس له اسم	لها اسم
نقوم بعمل Compile كل مره يُنفذ .	نقوم بعمل Compile له مره واحده فقط .
لا يُخزن في الـ Database	يُخزن في الـ Database
لا نستطيع تنفيذه من App اخري	نستطيع تنفيذه من App اخري
لا يأخذ Parameters	ممكن ان يأخذ Parameters

Syntax

```
Create [ or Replace ] Procedure Procedure_Name
[ (Parameter1 [Mode1] Datatype1, Parameter2 [Mode2] Datatype2, ... ) ]
Is | As
    Variables;
BEGIN
END [Procedure_Name];
```

ملحوظة:-

- ◀ ونقوم بإستخدام كلمة Replace للتعديل علي الـ Procedure .
- ◀ الـ Procedure تستطيع ان تأخذ Parameters او لا .
- ◀ عند عمل الـ Procedure او الـ Function فإن الـ Compiler يقوم بعمل Check عليهم فاذا كان الـ Syntax ليس به اخطاء فانه يظهر رساله تقول Procedure Created واذا كان هناك Error فسوف تظهر هذه الرساله Warning: Function Created with Compilation Error .
- ◀ لمعرفة الخطأ الموجود نقوم بكتابه Show Errors .

■ قم بعمل Procedure تقوم بإضافه اداره جديده .

```
CREATE PROCEDURE Add_Dept
IS
    V_Dept_Id      Number := 280 ;
    V_Dept_Name    Varchar2(20) := 'Outbox';
BEGIN
    Insert into Departments ( Department_Id,Department_Name)
    Values (V_Dept_Id , V_Dept_Name );
    DBMS_OUTPUT.PUT_LINE ('Inserted '|| SQL%Rowcount ||' row ');
END;
```

بهذا الشكل قد قمنا بعمل Procedure ولكنها لم تُنفذ بعد وان اردنا ان ننفذها فان هناك طريقتين وهما :-

1 Execute Add_Dept ;

او

2 BEGIN
Add_Dept ;
END;

ملحوظة:-

- ◀ يُفضل استخدام الطريقه الثانيه لان الطرق الاخرى لا تعمل في الفورمز مستقبلا .

Parameters

وهي شبيهة بالمتغيرات ولكنها تمرر القيمة فقط داخلها اما المتغير فانه يقوم بحمل القيمة ونقوم بتعريف الـ Parameter بعد اسم الـ Procedure ولا يأخذ Length

P_Name Varchar2(20)	×	P_Name Varchar2	✓
---------------------	---	-----------------	---

والـ Parameter له انواع وهي :-

IN	وهو الـ Default ونقوم بتمرير قيم من خارج الـ Procedure الي داخلها حتي نقوم بتنفيذ العملية اي ان الـ Pro تنتظر مني قيمة حتي تنفذ العملية .
OUT	تقوم الـ Procedure بتنفيذ العملية المطلوبة وتمرر الناتج من خلال الـ Parameter الي المتغير اي ان الـ Pro هي التي ستعطيني قيمة ولذلك لابد من عمل متغير في هذه الحالة
IN OUT	نقوم بإعطاء الـ Procedure قيمة ثم نقوم هي بعمل عملية علي هذه القيمة ومن ثم تخرجها لنا بشكل اخر.

ملحوظة :-

Formal Parameters < وهو الـ Parameter في حاله الانشاء مثل :-

Create Procedure Add_Name (P_Name Varchar2)

Actual Parameters < وهو الـ Parameter في حاله اعطائه القيمة مثل :-

```
BEGIN
  Add_Name (' Ahmed Habib');
END;
```

Using IN Parameters: Example

- قم بعمل **Procedure** نقوم بإعطائها رقم الموظف ونسبه الحوافز وتقوم هي بالتعديل في مرتبه بهذه النسبه .

```
CREATE OR REPLACE PROCEDURE Raise_Salary
(P_Id IN Employees.Employee_Id%Type, P_Percent IN Number)
IS
BEGIN
    Update Employees
    Set Salary = Salary + (Salary * P_Percent)
    Where Employee_Id = P_Id;
END Raise_Salary;
```

ولتنفيذها

```
BEGIN
    Raise_Salary ( 100 , .4 );
END;
```

Using OUT Parameters: Example

- قم بعمل **Procedure** نقوم بإعطائها رقم الموظف وهي تعطينا اسمه ومرتبه .

```
CREATE OR REPLACE PROCEDURE Query_Emp
(P_Id IN Number ,P_Name OUT Varchar2, P_Sal OUT Number)
IS
BEGIN
    Select Last_Name, Salary Into P_Name, P_Sal
    From Employees
    Where Employee_Id = P_Id;
END Query_Emp;
```

ولتنفيذها

```
DECLARE
    V_Name Varchar2(20);
    V_Sal Number;
BEGIN
    Query_Emp ( 171 , V_Name , V_Sal );
    DBMS_OUTPUT.PUT_LINE ( V_Name || ' ' || V_Sal );
END;
```

طريقة اخرى للتنفيذ

```
VARIABLE V_Name Varchar2(25)
VARIABLE V_Sal Number
EXECUTE Query_Emp (171, :V_Name , :V_Sal) ;
/
PRINT V_Name , V_Sal
```

ملحوظة:-

◀ لابد من عمل Variable عندما نقوم بالنداء علي Procedure بها
Parameter من النوع Out .

Using IN OUT Parameters: Example

■ سوف اعطيك رقم موبيل واريد استخراجه بهذا الشكل 010-0984-455.

```
CREATE OR REPLACE PROCEDURE Format_Phone
(P_Phone_No IN OUT Varchar2)
IS
BEGIN
P_Phone_No := '(' || Substr (P_Phone_No,1,3) ||
                ')' || Substr (P_Phone_No,4,3) ||
                '-' || Substr (P_Phone_No,7);
END ;
```

ولتنفيذها

```
DECLARE
V_Phone Varchar2(20) := '01009844556';
BEGIN
Format_Phone (V_Phone );
DBMS_OUTPUT.PUT_LINE (V_Phone );
END;
```

ملحوظة:-

◀ لابد من عمل Variable عندما نقوم بالنداء علي Procedure بها
Parameter من النوع In Out .

Syntax for Passing Parameters

نستطيع النداء علي الـ Parameters بأكثر من شكل :-

<u>Positional</u>	(الترتيب) نقوم بوضع قيم الـ Parameter بنفس الترتيب الموجود في الـ Procedure .
<u>Named</u>	(الاسم) نقوم بكتابه اسم الـ Parameter وبجواره القيمه الخاصه به . P_Name => 'Ahmed Habib'
<u>Combination</u>	(الاسم والترتيب) نقوم بكتابه القيم بالطريقتين السابقتين معا .

■ قم بإنشاء Sequence تبدأ من 1000 وتزيد بمعدل 1.

Create Sequence Habib_Seq Start With 1000 Increment By 1 ;

■ قم بعمل Procedure نقوم بإعطائها رقم الموظف وهي تعطينا اسمه ومرتبته ثم قم بالنداء عليها بالثلاث اشكال السابقه .

```
CREATE OR REPLACE PROCEDURE Add_Dept
( P_Name Varchar2, P_Loc IN Number )
IS
BEGIN
    Insert Into Departments ( Department_Id, Department_Name,
                             Location_Id)
    Values (Habib_Seq.Nextval , P_Name, P_Loc);
END Add_Dept;
```

للتنفيذ

- **Passing by Positional Notation**
Execute Add_Dept ('Education' , 2300);
- **Passing by Named Notation**
Execute Add_Dept (P_Loc => 2400, P_Name => 'Training');
- **Passing by Combination Notation**
Execute Add_Dept ('Advertising' , P_Loc => 2500);

Invoking Procedures

ونستطيع النداء علي ال-Parameters من خلال anonymous blocks وتعرضنا لامثله كثيره في هذا الصدد و ايضا من خلال Procedure اخري .

CREATE OR REPLACE PROCEDURE Process_Employees
IS

Cursor Emp_Cursor is Select Employee_Id
From Employees;

BEGIN

For Rec in Emp_Cursor Loop
Raise_Salary Rec.Employee_Id , .3);
End Loop;
Commit;

END Process_Employees;

Handled Exceptions

علي اقتراض اننا قمنا بعمل Procedure تُسمي Add_Dept وقمنا بعمل Exception لها من الاخطاء التي ممكن ان تقع بها ثم قمنا بعمل Procedure اخري ووظيفتها هي النداء علي Add_Dept .

■ فماذا سوف يحدث لو حدث خطأ في ال-Add_Dept ولم نكن قمنا بمعالجته ؟

اذا كنا قد قمنا بمعالجته فلن يحدث شيئاً اما اذا كان هناك Error ولو يعالج فسوف يظهر لنا Error ويقف كل شيء .

Removing Procedures

Syntax

DROP Procedure Procedure_Name ;

Example

DROP Procedure Raise_Salary;

Viewing Procedures in the Data Dictionary

■ لرؤيه ومعرفه الكود الذي قمت بكتابته داخل الـ Procedure التي قمت بعملها

```
Select *  
From User_Source  
Where Name = 'Add_Dept '  
And Type = 'PROCEDURE'  
Order by Line;
```

■ لرؤيه ومعرفه اسماء الـ Procedure التي قمت بعملها .

```
Select Object_Name  
From User_Objects  
Where Object_Type = 'PROCEDURE';
```

OUT BOX

CHAPTER 2

Creating Stored Functions

وهو نوع من انواع البلوكات ويعتبر احد افراد عائلته الـ Subprogram وتكوينه شبيهه بالبلوك العادي الـ Anonymous مع اختلاف بسيط ؛ لا نستخدم Declare مع الـ Function ونستخدم بدلا منها Is او As ولا بد من كتابه Is او As حتي لو لم نضطر الي تعريف Variables اما والباقي كما هو . وميزه الـ Function اننا نقوم بكتابه الكود مره واحده ونستطيع تنفيذها والنداء عليها اكثر من مره لانها مُخزنه في الداتا بيز ولا بد ان تعود الـ Function بقيمه .

ما الفرق بين الـ Subprogram (Function , Procedure) و الـ Anonymous Block

<u>Anonymous</u>	<u>Subprogram</u>
ليس له اسم	لها اسم
نقوم بعمل Compile كل مره يُنفذ .	نقوم بعمل Compile له مره واحده فقط .
لا يُخزن في الـ Database	يُخزن في الـ Database
لا نستطيع تنفيذه من App اخري	نستطيع تنفيذه من App اخري
لا يأخذ Parameters	ممكن ان يأخذ Parameters

Syntax

```
CREATE [ OR REPLACE ] FUNCTION Function_Name
    [(Parameter1 [mode1] Datatype1, .....)]
RETURN Datatype
IS|AS
    [Variables ; ...]
BEGIN
RETURN expression;
END [function_name];
```

ملحوظة:-

- ◀ الفرق بين الـ Procedure والـ Function في Return Datatype اي اننا نقوم بتهيئة الـ Function ان الناتج المراد استخراجه سيكون من النوع الذي قمت بتحديدته ونستقبله من خلال Return Expression الذي فيه نحدد المتغير الذي نريد ان نضع فيه القيمة .
- ◀ لا نستطيع استخدام Parameter الا من النوع In فقط .
- ◀ الـ Parameters هي كما عرفناها سابقا وعند تعريفها فاننا لا نعطيها Length ووايضا الـ Return عند تحديد الـ Datatype لا نحدد لها Length ايضا .

■ قم بعمل Function نعطيه رقم الموظف وتعطينا مرتبه .

```
CREATE OR REPLACE FUNCTION Get_Sal ( P_Id Number )
RETURN Number
IS
V_Sal Number:= 0;
BEGIN
Select Salary into V_Sal
From Employees
Where Employee_id = P_Id;
RETURN V_Sal;
END Get_Sal;
```

ولتنفيذها (هناك اربعة طرق):-

1	Execute DBMS_OUTPUT.PUT_LINE (Get_Sal(100));
2	Variable V_Sal Execute :V_Sal := Get_Sal(100);
3	DECLARE V_Sal Number ; BEGIN V_Sal := Get_Sal(100) ; DBMS_OUTPUT.PUT_LINE (v_Sal); END;
4	Select Job_Id , Get_Sal (100) From Employees;

ملحوظة :-

- ◀ لابد من عمل متغير عند النداء علي الـ Fun لاسترجاع القيمة فيه .

- قم بعمل Function نعطيها مرتب وتعطينا صافي الضريبة الخاص به .

```
CREATE OR REPLACE FUNCTION Tax (P_Value IN Number)
RETURN Number
IS
BEGIN
RETURN (P_Value * 0.08);
END Tax;
```

ولتنفيذها

```
Select Employee_Id, Last_Name, Salary, Tax(Salary)
From Employees
Where Department_Id = 100;
```

- قم بعمل Function نعطيها رقم موظف وتعود بـ True لو مرتبه اكبر من متوسط مرتبات ادارته و False لو العكس .

```
CREATE FUNCTION Check_Sal ( P_Id Number)
RETURN Boolean
IS
V_Dept_Id Employees.Department_Id%Type;
V_Empno Employees.Employee_Id%Type ;
V_Sal Employees.Salary%Type;
V_Avg_Sal Employees.Salary%Type;

BEGIN
Select Salary,Department_Id into V_Sal,V_Dept_Id
From Employees
Where Employee_Id = P_Id;
Select Avg(Salary) into V_Avg_Sal
From Employees
Where Department_Id=V_Dept_Id;
IF V_Sal > V_Avg_Sal THEN
RETURN True;
ELSE RETURN False;
End IF;
EXCEPTION
When NO_DATA_FOUND Then RETURN Null;
END;
```

ولتنفيذها

```
DECLARE
  V_Check Boolean ;
BEGIN
  V_Check := Check_Sal ( 205) ;
  IF      V_Check = True  Then
    DBMS_OUTPUT.PUT_LINE ('T');
  ELSIF V_Check = False  Then
    DBMS_OUTPUT.PUT_LINE ('F');
  End IF ;
END;
```

ملحوظة:-

◀ لرؤيه الخطأ نكتب Show Errors .

Removing Functions

Syntax

```
DROP Function Function_Name ;
```

Example

```
DROP Function Get_Sal ;
```

Viewing Functions in the Data Dictionary

▪ لرؤيه ومعرفه الكود الذي قمت بكتابته داخل الـ **Function** التي قمت بعملها

```
Select Text
From User_Source
Where Name = ' Get_Sal '
And Type = 'FUNCTION' ;
```

▪ لرؤيه ومعرفه اسماء الـ **Functions** التي قمت بعملها .

```
Select Object_Name
From User_Objects
Where Object_Type ='FUNCTION';
```

OUT BOX

تابعوا جديد عروضنا وخدماتنا عبر بواباتنا الاجتماعية



OUT BOX

Effective Social World

 **outboxeg**

 **@outboxeg**

www.outbox-eg.com

 **0502212232**  **01140000286**

تم بحمد الله تعالى

لا تنسونا بالدعاء

Ahmed Habib

Email: Dev_Habib@yahoo.com

Mobil: 002 0100 9844556

Oracle Developer
Instructor Oracle Developer
Oracle Financial Consultant
Instructor Oracle Financial R12

والى لقاء قريب فى الكتاب الثانى فى الـ PL/SQL فىما يتعلق بالـ Package و الـ Trigger.